A FAST PARALLEL SELECTION ALGORITHM ON GPUs

Darius Bakunas-Milanowski[1], Vernon Rego[2], Janche Sang[1], and Chansu Yu[1]
[1]Dept. of Electrical Engineering and Computer Science
Washkewicz College of Engineering
Cleveland State University, Cleveland, OH, USA
[2] Dept. of Computer Science
Purdue University, West Lafayette, IN, USA

ABSTRACT:
The computing power of current Graphical Processing Units (GPUs) has increased rapidly over the years. They offer much more computational power than recent CPUs by providing a vast number of simple, data parallel, multi-threaded cores. In this paper, we propose a new parallel selection algorithm and compare the performance of different parallel selection algorithms on the current generation of NVIDIA GPUs. That is, given a massively large array of elements, we were interested in how we could use a GPU to efficiently select those elements that meet certain criteria and then store them into a target array for further processing. The optimization techniques used and implementation issues encountered are discussed in detail. Furthermore, the experimental results show that our implementation performs an average of 3.67 times faster than Thrust, an open-source parallel algorithms library.

KEYWORDS:
Parallel Selection, CUDA, Thrust Library, GPU, Optimization Techniques, SIMT

1. Introduction

Selection, also known as stream compaction or filtering, is a common programming concept that has a wide range of applications in the area of statistics, database software, artificial intelligence, image processing, and simulations [1] [2][3]. It produces a smaller output array, containing only wanted elements from the input array made up of the mixed elements. With the tremendous amount of data elements to be processed, better performance becomes a key factor in implementing these algorithms. Therefore, exploiting the availability and the power of multiprocessors to speed up the execution is of considerable interest.

In the past few years, modern Graphics Processing Units (GPUs) have been increasingly used together with CPUs to accelerate a broad array of scientific computations in so-called heterogeneous computing [4]. It is now much more convenient to create application software that will run on current GPUs for processing massively large amounts of data, without the need to write low-level assembly language code. Furthermore, a selection of accelerated, high performance libraries allows an easy way of adding GPU-acceleration to the wide array of scientific applications. One can get even more flexibility and speed by writing his or her own GPU-accelerated programs using the CUDA Thrust Library [5], which provides a comprehensive development environment for C and C++ developers.

NVIDIA Kepler GPUs (Figure 1) consist of a scalable number of streaming multiprocessors (SMXs), each containing a group of streaming processors (SPs) or cores to execute the light-weighted threads, warp by warp, using the Single Instruction, Multiple Threads (SIMT) style (term coined by NVIDIA manufacturer). In addition to the main memory on the CPU motherboard, the GPU device has its own off-chip device memory (i.e. global memory). The kernel function, which is executed on the device, is composed of a grid of threads. Note that a grid is divided into a set of blocks and each block contains multiple warps of threads. Blocks are distributed evenly to the different SMXs to run. Furthermore, registers and shared memory in a SMX are on-chip memory and can be accessed very fast. They are per-block resources and are not released until all the threads in the block finish execution. Each SMX also has 32 special function units (SFUs) for fast approximate transcendental operations, like __cosf(), __expf(), etc. and 32 load/store (LD/ST) units for memory read/write operations.



Figure 1: *Kepler GPU block diagram*

In this paper, we focused on the design and implementation of a new parallel selection algorithm and compared its performance with other parallel selection methods on CUDA-enabled GPUs. All tests were performed using CUDA Toolkit on a PC with a consumer grade NVIDIA GeForce GTX 770 GPU and also on the Ohio Supercomputer Center's newest cluster Ruby, outfitted with professional NVIDIA Tesla K40 GPUs. Both of these cards belong to the NVIDIA® Kepler [6], a cutting-edge high performance computing architecture. The empirical results show that our algorithm, which also preserve the relative order of the input elements, performs much faster than the Thrust library and also slightly outperforms referenced Adinetz version.

The organization of this paper is as follows. Section 2 describes related work. Section 3 goes in to details of our implementation and finally, in Section 4, the experiments and the results for performance evaluation are presented. We give a short conclusion in Section 5.



Figure 2: *Selection of data elements using (a) atomicInc (b) list ranking*

2. Related Work

Sequential selection is a common function and it is available in many programming languages and/or libraries. However, to implement the parallel selection, the challenging is how to determine the indices of the selected elements in the destination array. In general, the approaches to performing the stream compaction on multiprocessors can be classified into two categories: one is based on the atomic operation, while the other is based on the list ranking using the prefix-sum algorithm [7], as described below in detail.

2.1 Atomic operation based approaches

In the former approach, we can use an index counter, which will be incremented by one for each newly selected element. Since many threads share the counter, the addition has to be an atomic operation. This can be done by using CUDA atomicInc() function, as illustrated in Figure 2a. Note that a CUDA atomic function performs a read-modify-write atomic operation on one 32-bit or 64-bit word residing in global or shared memory.

The main problem with this approach is that these atomic operations become a major bottleneck when the input contains a large amount of elements that pass our selection criteria. This is due to the very large number of threads competing to increment the single counter inside the global memory.

One possible improvement is to use shared memory atomics. This will essentially decrease the number of atomic collisions to a block size. Unfortunately, its performance still suffers from the thread synchronization. As demonstrated in the Experimental Results Section, execution time for both algorithms is directly proportional to the number of passing items.

A modified approach is discussed in the article "CUDA Pro Tip: Optimized Filtering with Warp-Aggregated Atomics" [8], written by researcher A. Adinetz. In addition to using the aggregated atomicAdd(), it uses the primitives ballot(), __ffs() and __popc() (Compute Capability 2.0 and above) to perform intra-warp scan [9] and also uses the warp shuffle intrinsic [10] which is available on the Kepler and later GPUs (Compute Capability 3.0 and above) to broadcast the group index value to all of the threads within a warp. His implementation of atomic function with warp aggregation is isolated from the rest of the application, and can be used as a drop-in replacement for existing code that use CUDA atomics. Furthermore, since each warp will issue at most one atomicAdd() request, its execution time will not be proportional to the number of passing elements and hence can be reduced greatly.

Note that these three algorithms based on atomic operation do not preserve the relative order of the input elements, thus might not be suitable for certain applications.

2.2 List ranking based approaches

For the latter approach, one such implementation is provided by the Thrust library, specifically a method called copy_if() [11], which is a fairly good implementation, simple to use and may prove to be the best choice for most users needing this operation. By digging into its implementation details, we found that it uses 2-level sums. First, it calculates the number of selected elements within the block (using index counter inside the shared memory) and stores the result in an intermediate array of size N / block_size (N is the total number of input elements) in the global memory. Then it performs a parallel prefix sum (Figure 2b) on this array and uses the outcome in the final phase to determine the indices of output elements. As a result, the relative order of the input elements is also preserved. Furthermore, as shown in the later section, Thrust algorithm execution time does not depend on the number of passing elements.

Our previous work in [12] is also based on the list-ranking approach. It used the _ballot() and _popc() to find the offset of a selected element in a warp quickly. We also optimized the code to let each thread handle many elements to increase the computation/communication granularity. It performs much faster than the Thrust library, but slower than the Adinetz algorithm.

3. A New Algorithm



Figure 3: *Phase 1a: fill predicate and counter arrays*

The following algorithm implementation consists of three major phases. The phase 1 kernel (Figure 3) starts by evaluating a predicate for each subgroup of 32 elements and saving the result into a predicate array. The number of passing elements in the subgroup is also determined (using __popc() instruction) and saved inside the register variable cnt for each thread in the warp. This operation is performed for 32 iterations. As a result, each warp processes the total number of 1024 elements. When this loop completes, a parallel reduction operation is performed on the cnt register values, resulting in the number of selected elements for each 1024-element group. Note that __shfl_down() instruction was used, that essentially allows passing down register values from the lane with higher ID relative to the caller lane (Figure 5). The result is saved in the counter array.

In phase 2, a prefix sum operation is applied to the counter array (Figure 6). As a result, there are counter[k – 1] valid elements before the group k. Note that for this operation we simply used Thrust implementation thrust::inclusive_scan(x), which was fast enough and sufficient for our purposes.

```
1  #include <cuda.h>
2  #include "algorithms.h"
3  #include "err.h"
4
5  #define WARP_SZ 32
6
7  __device__ inline int lane_id(void) { return threadIdx.x % WARP_SZ; }
8
9  __global__ void alg7_kernel_phase1(float *input,
10   unsigned int *counter,
11   unsigned int *pred,
12   float percent, const
13   unsigned int num_items)
14 {
15   int tid = blockIdx.x * blockDim.x + threadIdx.x;
16
17   if (tid >= (num_items >> 5) )  // divide by 32
18     return;
19
20   int lnid = lane_id();
21   int warp_id = tid >> 5; // global warp number
22
23   unsigned int mask;
24   int cnt;
25
26   for(int i = 0; i < 32 ; i++) {
27       mask = __ballot(input[(warp_id<<10)+(i<<5)+lnid] <= percent);
28
29       if (lnid == 0)
30           pred[(warp_id<<5)+i] = mask;
31
32       if (lnid == i)
33           cnt = __popc(mask);
34   }
35   // para reduction to a sum of 1024 elements
36 #pragma unroll
37   for (int offset = 16 ; offset > 0; offset >>= 1)
38       cnt += __shfl_down(cnt, offset);
39
40   if (lnid == 0)
41       counter[warp_id] = cnt; // store the sum of the group
42 }
43
```

Figure 4: *Phase 1 kernel*



Figure 5: *Phase1b: perform reduction operation on cnt values*

Figure 6: *Phase 2: perform inclusive scan operation on the counter array*

Phase 3 produces the final result. The process begins by reading the predicate array that was produced in the phase 1. The number of set bits is determined (using __popc() instruction) for each predicate value and saved into the cnt register for each lane inside the warp (Figure 7).



Figure 7: *Phase 3a: get bit counts for the predicate values*



Figure 8: *Phase 3b: perform prefix sum on cnt values*

```
44 /* PHASE3: produce final result array */
45 __global__ void alg7_kernel_phase3(int *output,
46 unsigned int *counter,
47 unsigned int *pred,
48 const unsigned int num_items){
49
50  int tid = blockIdx.x * blockDim.x + threadIdx.x;
51  if (tid >= (num_items >> 5) )  // divide by 32
52 return;
53
54  int lnid = lane_id();
55  int warp_id = tid >> 5; // global warp number
56
57  unsigned int predmask;
58  int cnt;
59
60  for(int i = 0; i < 32 ; i++) {
61      if (lnid == i) {
62          // each thread take turns to load its local var (i.e regs)
63          predmask = pred[(warp_id<<5)+i];
64          cnt = __popc(predmask);
65      }
66  }
67 // parallel prefix sum
68
69 #pragma unroll
70  for (int offset=1; offset<32; offset<<=1) {
71      int n = __shfl_up(cnt, offset) ;
72      if (lnid >= offset) cnt += n;
73  }
74
75  int global_index =0 ;
76  if (warp_id > 0)
77      global_index = counter[warp_id -1];
78
79  for(int i = 0; i < 32 ; i++) {
80      int mask = __shfl(predmask, i); // broadcast from thr i
81      int sub_group_index = 0;
82      if (i > 0)
83          sub_group_index = __shfl(cnt, i-1); // broadcast from thr i-1 if i>0
84
85      if (mask & (1 << lnid ) ) // each thread extracts its pred bit
86        output[global_index + sub_group_index +
87          __popc(mask & ((1 << lnid) - 1))] = (warp_id<<10)+ (i<<5) + lnid  ;
88  }
89 }
```

Figure 9: *Phase 3 kernel*

4. Experimental Results

We compared our implementation with the Thrust copy_if() method and also the Adinetz version. The following experiments were conducted on one of the nodes in the Ruby cluster provided by the Ohio Supercomputer Center. The GPU used in this particular computing platform was the NVIDIA Tesla K40m, which contains 15 multiprocessors (2880 CUDA cores in total) and 12GB GDDR5 memory. A warp, the scheduling unit in CUDA, has 32 threads that perform SIMT computation on a multiprocessor. The device programs use a CUDA compiler driver 7.0. In the first experiment, we measured the execution times for all of the algorithms by varying the number of threads per block. This allowed us to select the optimal kernel configuration for our future experiments (Figure 11). An interesting point is that performance for the global counter version does not depend on the kernel configuration, while the algorithm using shared memory could be further optimized by choosing appropriate block size – we found that 128 threads per block work best for Tesla K40m device. This also proved to be the case for the advanced algorithm.



Figure 11: *Performance metrics for various kernel block dimensions tested (N = 67,108,864)*

We measured the execution performance of these algorithms by varying the number of items being selected (Figure 12). We chose the block size of 128 threads. Note, for this experiment we used uniform random distribution for our data source. It can be seen that the atomic operation based approaches using single global counter and counter in the shared memory can perform better than the Thrust library only when the percentage p is very small because their execution times are proportional to the number of the selected elements. The running times of the Thrust copy_if(), Adinetz' and our both methods remain almost unchanged because they are independent of the percentage p (i.e. the number of valid elements).

We also measured the execution times for each three phases of our improved algorithm, as illustrated in Figure 13. Phases 1 and 3 were the biggest contributors to overall performance of the algorithm, which is also why we chose not to implement the inclusive scan operation for phase 2 ourselves.



Figure 12: *Performance comparison with different number of selected items (uniform random distribution)*

For our final experiment, we measured algorithm performance based on the input size. As figure 14 shows, the performance of all tested algorithms was directly proportional to the number of elements while the Trust implementation was affected to a higher degree.

5. Conclusion

Our work represents an advanced implementation of a parallel selection algorithm. The experiment results are encouraging, as we were able to achieve 3.67 times better performance than what is possible using Thrust implementation. Furthermore, our new algorithm not only outperforms the Adinetz's version, but also preserves the order of the selected elements and this feature, we believe, is more important for most current applications.



Figure 13: *Execution time breakdown for our algorithm*



Figure 14: *Performance comparison with different number of elements in the source array (p = 50%)*

Acknowledgments

This research was supported by Cleveland State University 2015 Undergraduate Summer Research Award Program and by allocation of computing time from the Ohio Supercomputer Center.

References

[1] D. B. Kirk andW.-m.W. Hwu, ProgrammingMassively Parallel Processors: A Hands-on Approach, 1st ed. San Francisco, CA, USA: Morgan Kaufmann Publishers Inc., 2010.

[2] J. Hoberock and N. Bell, "Thrust: A parallel algorithms library which resembles the C++ Standard Template Library (STL)," 2015. [Online]. Available: http://thrust.github.io

[3] G. Diamos, H. Wu, A. Lele, J. Wang, and S. Yalamanchili. Efficient relational algebra algorithms and data structures for GPU. Technical Report GIT-CERCS-12-01, CERCS, Georgia Institute of Technology, 2012

[4] S.-H. Lo, C.-R. Lee, I.-H. Chung, and Y.-C. Chung, "Optimizing Pairwise Box Intersection Checking on GPUs for Large-Scale Simulations," ACM Trans. on Modeling and Computer Simulation, vol. 23, no. 3, pp. 19:1–19:22, July 2013.

[5] J. Sang, C. Lee, V. Rego, and C. King, "A fast implementation of parallel discrete-event simulation on GPGPU," in Proceedings of Int'l Conference on Parallel and Distributed Processing Techniques and Applications, 2013.

[6] Nvidia.com, "NVIDIA Kepler Compute Architecture | High Performance Computing | NVIDIA," 2015. Web. 2 Aug. 2015. [Online]. Available: http://www.nvidia.com/object/nvidia-kepler.html

[7] J. C. Wyllie, "The complexity of parallel computations," PhD thesis, Cornell University, Ithaca, NY, USA, Tech. Rep., 1979.

[8] A. V. Adinetz. 'CUDA Pro Tip: Optimized Filtering With Warp-Aggregated Atomics'. Parallel Forall. N.p., 2014. Web. 3 Aug. 2015. [Online]. Available: http://devblogs.nvidia.com/parallelforall/cuda-pro-tip-optimized-filtering-warp-aggregated-atomics/

[9] M. Harris and M. Garland. "Optimizing Parallel Prefix Operations for the Fermi Architecture", Chapter 3 of the book "GPU Computing Gems - Jade Edition", Morgan Kaufmann Publishers Inc., October 2011

[10] M. Harris. 'CUDA Pro Tip: Do The Kepler Shuffle'. Parallel Forall. N.p., 2014. Web. 3 Aug. 2015. [Online]. Available: http://devblogs.nvidia.com/parallelforall/cuda-pro-tip-keplesr-shuffle/

[11] J. Hoberock and N. Bell, "Stream Compaction," 2015. [Online]. Available: https://thrust.github.io/doc/group__stream__compaction.html

[12] D. Bakunas-Milanowski, V. Rego, J. Sang, C. Yu, "An improved implementation of parallel seclection on GPUs", accepted to appear in the International Symposium on Software Engineering and Applications, 2015